

Predicting the Orbital Period of Exoplanets by Analyzing the Wobble of Stars

▼ Introduction

Stars are pulled in a circle or ellipse in response to the gravity of orbiting planets. By analyzing the "wobble" (or radial velocity) of a star, astronomers can predict the presence and orbital period of exoplanets.

Radial velocity is recorded, often over months or years, with a spectrograph connected to a telescope. This data is used to generate a periodogram, in which a peak is evidence of an exoplanet; the orbital period of the exoplanet is given by the location of the peak.

However cloud cover, scheduling conflicts and other issues can often disrupt observations, so data is generally not regularly sampled. This means that standard Fourier techniques cannot be used to generate a periodogram, and other approaches are needed. A common method for the frequency analysis of irregularly sampled data is the Lomb-Scargle technique.

Fischer (2003) recorded the radial velocity of the star HD 3561 (also known as [54 Piscium](#)), and presented a periodogram to show evidence of an exoplanet (now known as [54 Piscium b](#)).

This application uses Maple 2019's new [Lomb-Scargle tools](#) to reproduce the analysis; the periodogram shows a periodicity of 62.2 days, agreeing with value given by Fischer (2003).

References:

- [A Sub-Saturn Mass Planet Orbiting HD 3651, Fischer D.A. et al., The Astrophysical Journal, 590:1081–1087, 2003 June 20](#)
- Radial velocity data found at http://astrostatistics.psu.edu/datasets/exoplanet_Doppler.html

```
> restart:  
with(ColorTools):  
with(SignalProcessing):
```

▼ Import and Plot Radial Velocity Data

```
> data := ImportMatrix("this://HD 3651.csv")
```

	"JD-2440000"	"Velocity (m/s)"	"Uncertainty (m/s)"	"Observatory"
<i>data</i> :=	7047.854	13.74	7.88	"Lick"
	7374.94	-3.19	7.15	"Lick"
	7430.758	9.3	6.65	"Lick"
	7846.849	-7.82	8.95	"Lick"
	8113.959	-4.21	9.31	"Lick"
	8199.717	-13.79	8.41	"Lick"
	8645.68	-3.08	10.52	"Lick"
	8845.939	15.77	7.16	"Lick"
	8846.983	26.93	6.16	"Lick"
	⋮	⋮	⋮	⋮


139 × 4 Matrix

(2.1)

```
> timeData := data[2.., 1]:
   radialVelocityData := data[2.., 2]:

plot(
  timeData, radialVelocityData
  ,style      = pointline, symbol = solidcircle, color = Color
  ("RGB", [0, 79/255, 121/255])
  ,axes      = boxed, size = [800, 400]
  ,labels    = ["Julian Date - 2440000", "Radial Velocity (m/s)
  "], labelfont = [Calibri], labeldirections = [horizontal,
  vertical]
  ,axesfont  = [Calibri]
  ,title     = "Radial Velocity of HD 3651", titlefont =
  [Calibri, 16]
  ,background = Color("RGB", [218/255, 223/255, 225/255])
  ,axis      = [gridlines = [color = Color("RGB", [1, 1, 1])]]
  );
```

Radial Velocity of HD 3651




▼ Lomb Scargle Periodogram

Calculate and plot spectra

```
> frequencies, power := LSSpectrum(timeData, radialVelocityData,  
  minimumfrequency = 1/10000, maximumfrequency = 0.1, outputdata =  
  5000):
```

```
plot(  
  1/~frequencies, power  
  ,thickness = 0, color = Color("RGB",[0, 79/255, 121/255])  
  ,axes = boxed, size=[800, 400], axesfont = [Calibri]  
  ,labels = ["Period (days)", "Power"], labelfont = [Calibri],  
  labeldirections = [horizontal, vertical]  
  ,title = "Lomb-Scargle Periodogram of the Radial Velocity  
of HD 3651", titlefont = [Calibri,16]  
  ,background = Color("RGB", [218/255, 223/255, 225/255])  
  ,axis[1] = [mode = log, gridlines = [color = Color("RGB", [1,  
1, 1]]]  
  ,axis[2] = [gridlines = [color = Color("RGB", [1, 1, 1]]]  
);
```

Lomb-Scargle Periodogram of the Radial Velocity of HD 3651



Note the peak - its presence is evidence of an exoplanet. Hence the orbital period (in days) of the exoplanet is

```
> maxPowerIndex := max[index] (power) :  
  1/frequencies [maxPowerIndex] ;  
  62.239
```

(3.1)

This matches the value given by Fischer (2003), and also stated [here](#).